

第五章 CE—150打印机的使用及有关的语句

SHARP PC—1500计算机系统中，CE—150打印机是重要的组成部分，它除具有打印输出功能以外，还可通过它加接磁带机、扩展接口、X—Y坐标绘图机等外部设备。在这一章里，主要介绍CE—150打印机及其有关的打印命令的情况。有关磁带机的连接使用将在第六章中专门介绍。

CE—150打印机与主机PC—1500的连接安装、装纸换笔、充电等各项机器本身的准备操作注意事项，已在第一章中介绍机器时已作了介绍，在这一章中就不复述了。

CE—150打印机面板上有一“PRINT”开关，这是控制键盘运算是否打印的开关。当开关拨到“P”一侧时，我们在RUN方式之下所作的常规键盘运算与保留函数运算，其运算式和结果均由打印机打印出来。如果开关拨到“·”一侧，CE—150打印机对键盘运算不予打印，只能在显示屏上显示。

本章中介绍的各种打印、绘图的语句命令的执行，是不受“PRINT”开关控制的，不管是在键盘上直接操作执行也好，放在程序之中作为语句形式运行也好，均可各自根据其功能指挥打印机工作。

第一节 打印语句及指令

一、机器工作前的检查试验命令（TEST指令）

机器安装就绪后，可在键盘上输入TEST指令来检查。此项指令的输入，可在PRO或RUN之任一方式之下进行。

如果机器一切正常，则打印机印出四个彩色小方框，尺寸是5×5mm，颜色顺序应是黑、蓝、绿、红。

如果机器不正常，则显示相应的错误信息。此时再对机器进行必要的调整，以使其工作正常。

二、打印机工作方式选择及其命令（TEXT与GRAPH）

打印机的工作方式，可分为TEXT方式（文本方式）——打印数字，字符；GRAPH方式（绘图方式）——描绘图表，图形。

1、TEXT语句指令

形式：〈语句号〉TEXT

也可由键盘以键盘命令形式送入机器。

功能：命令机器采用TEXT方式，准备打印数字，字符。

在计算机工作时，主要或大量的输出，均是输出数字，字符，所以机器在关机后再开机，是自动接通TEXT方式的。

2、GRAPH语句指令

形式： 〈语句号〉GRAPH

也可由键盘以键盘命令形式送入机器。

功能：命令机器采用GRAPH方式，用直接或相对的 X—Y 坐标系统来描绘各种线条图形。

当GRAPH方式工作完毕，一般应用TEXT语句指令使机器恢复到TEXT方式。

这一节与下一节中介绍的有关打印与绘图的指令，都须注意是在什么工作方式（TEXT与GRAPH）之下使用的，因为有些命令只能在TEXT方式下执行，有些只能在GRAPH方式之下执行，搞错了机器显示[ERROR 73]。

三、字符、数字的打印指令

1、LRPRINT语句

LRPRINT语句在程序中用法与PRINT命令类似，LRPRINT是命令机器在打印机上输出，而不是在显示屏上输出。（有些不同的计算机LRPRINT与PRINT命令合为一，这要根据各自的规定来操作）。PRINT命令在执行时，机器正常停机并显示；而LRPRINT命令在打印完毕后，继续向下运行，而不需按执行键启动程序向下运行。

LRPRINT语句的一般形式：

形式1：〈语句号〉 LRPRINT 表达式

形式2：〈语句号〉 LRPRINT 表达式1〉，〈表达式2〉

形式3：〈语句号〉 LRPRINT 表达式1〉；〈表达式2〉；……；〈表达式n〉

LRPRINT的特殊形式：

〈语句号〉 LRPRINT

特殊形式表示在打印纸上空一行，也就是空行命令，除空行功能以外，它还有一“回车”功能。LRPRINT的打印形式可通过下面的例子来理解。

```
例(1) 10 A$="SHARP", B=123.45, F=487E 1
      20 LRPRINT 45678
      30 LRPRINT 45678+B
      40 LRPRINT "45678", A$
      50 LRPRINT "A$"; A$; F
      60 LRPRINT "B="; B; "A$="; A$; "F="; F
      70 LRPRINT A$; "□□□PC"; "-1500"
      80 END
```

LRPRINT执行时，有以下特点。

(1) 它在TEXT方式下执行时，打印字符是从左至右顺序打印的，一行最多可打印标准字符（即2号字体）18个，一行打印不下自动换行继续打印。

(2) 打印时不给定打印位置命令时，则数值打印在纸的右端，字符串或字符串变量的内容打印在纸的左端。

(3) 同PRINT一样，一条LRPRINT语句中只许用一逗号，而且若用了分号则不允许使用逗号。用了逗号后，逗号两边的内容分成两行打印。

(4) 分号的使用不受限制,只要一条LPRINT语句写得下即可。同PRINT类似,用分号分隔开的(表达式)的内容,以一个紧接一个的格式打印,并且中间不留空位。(注意不要将符号位错当成空格位。)一行打印不下时,自动换行打印。

(5) 打印字符的颜色与大小由下面所述的颜色命令与字体大小命令规定,如果不作规定则机器按标准字大小黑色来打印,或用机器曾规定过未予撤消的颜色与大小规定来打印。

在将LPRINT命令写入计算机时,与GPRINT类似,也不得在PRINT前加一个“L”,而必须一次写出,否则出现[ERROR 1]。

2、打印颜色指定语句与打印字符字体大小指定语句(COLOR与CSIZE)

(1) 颜色指定指令COLOR。在程序中一般形式为:

〈语句号〉 COLOR 〈a〉

其中〈a〉的值应为0~3的正整数。也可用变量来代替,但必须是已赋了值的变量,其值也应为0~3的正整数。〈a〉也可为〈表达式〉。

功能:通过COLOR命令指定打印字符或绘图时用何种颜色。PC-1500机规定、各颜色的编号为:

0 —— 黑色;	1 —— 蓝色;
2 —— 绿色;	3 —— 红色。

例如给定COLOR 2,则机器用绿色笔打印。

注意几点:

第一,为保证颜色命令的正确执行,在装色笔时必须按规定的顺序在笔架指定位置装色笔,如果色笔的位置装错,则COLOR命令执行也同样错。例如在应装红色笔的笔槽中装上了黑色笔,那么执行了COLOR 3命令后,用的笔是黑色(0号)而不是红色(3号)。换句话说,COLOR 3命令的执行,实际上是指令笔架转动的位置为3号位置,至于在3号笔架位置上装了什么颜色的笔,机器是不管的。

第二,当每次关机后再重新开机时,打印机都将调整笔架到COLOR 0位置,即黑色位置。

COLOR颜色指令可在TEXT与GRAPH方式下执行。也可作为键盘命令送入机器执行。

(2) 字符字体大小指定指令CSIZE

一般形式为:

〈语句号〉CSIZE a

其中〈a〉的值为1~9的正整数。若不为整数,机器自动取整,如果大于等于10,将产生ERROR 19。〈a〉也可是〈表达式〉,只是其值必须是在1~9的范围内。

功能:指定打印字符的字体大小。CSIZE后的数值,就是字体大小的号码,PC-1500机为用户准备有九种大小规格的字体。表5-1给出每种字体规格的大小尺寸。

因为常用的字体大小为2号字体,在开机后,机器自动准备的字体是2号字。因此我们把2号字称作标准字体。

CE-150打印机的纸宽是57mm,标准字体一行可打印18个字符。

CSIZE指令不仅可作为语句,放在程序中执行,还可作为键盘命令执行。但必须注意,机

字 符 字 体 大 小 表

表 5 — 1

CSIZE	1	2	3	4	5	6	7	8	9
打印一行字符数	36	18	12	9	7	6	5	4	4
字符高 (mm)	1.2	2.4	3.6	4.8	6.0	7.2	8.4	9.6	10.8
字符宽 (mm)	0.8	1.6	2.4	3.2	4.0	4.8	5.6	6.4	7.2

器规定CSIZE命令对于打印源程序指令LLIST，只能是CSIZE 1或CSIZE 2，其余的字体数均作2处理。如CSIZE 4命令或CSIZE 5等，机器仍当成CSIZE 2执行。

例（2）打印不同字体大小的程序

```

10 CLEAR
20 CSIZE 1
30 LPRINT "ABC895" : LPRINT
40 FOR A = 2 TO 9
50 CSIZE A
60 LPRINT "ABC748";
70 NEXT A
80 END
    
```

程序运行后，打印出如表 5 — 1 所列九种不同大小字体。

3、打印位置指定语句 (LCURSOR与TAB)

(1) LCUROR语句的一般形式：

〈语句号〉 LCURSOR (a)

其中：〈a〉为 0 ~35 的正整数。也可是〈表达式〉，其值也应为 0 ~35 范围内。

功能：给定打印起始位置，〈a〉是指定位置号。如：LCURSOR 0 表示从最左端开始打印，LCURSOR 5 表示从第 5 个字符位开始打印。

〈a〉的取值范围应根据选用字体大小而确定，也就是决定于 CSIZE 命令。因为 CE—150 打印机纸宽只有 57mm，不同大小的字体每行打印的字数不一（见表 5 — 1），最多为 36 个字符（1 号字），最少 4 个字符（9 号字）。所以在 CSIZE 1 命令时，最大取数为 35；而用 2 号字（标准字）打印时（即 CSIZE 2），最大取数为 17，如超过各种字体不同的数目的位置的数。则显示 [ERROR 19]。

LCURSOR 一般总是放在 LPRINT 语句之前，配合打印命令的。它只对一条紧跟在后的 LPRINT 语句有效。

LCURSOR 只能在 TEXT 方式中使用。

(2) TAB 函数指令

TAB 指令一般作为函数形式放在 LPRINT 语句之中，其一般形式为：

语句号 LPRINT TAB (a) ; 〈表达式〉

其中 TAB (a) 就是 TAB 函数形式。它给出的打印位置数指定在它之后的变量（或数）

的打印位置。TAB的功能与LCURSOR相同，只是它可放在LPRINT语句之中使用。

例如：300 LPRINT TAB 15; K

表示在15号位置打印变量K的值。

又如：250 LPRINT TAB 3; A; TAB10; F * H; TAB15; A \$

表示一行中，3号位置开始打印A的值，10号位程开始打印 F * H的值，15号位置开始打印A \$的内容。

TAB (a) 中的 (a) 的含义与取值，与LCURSOR语句完全一样，也是决定于符大小的。同样，它也只对一条紧跟它的打印命令有效。

TAB函数非常重要，用得很多，必须熟练掌握。

(3) USING格式在打印、语句中的应用

前面已介绍过自选格式函数语句USING在PRINT语句中的使用和用法。在LPRINT语句中，USING同样可以应用。它的格式，用法与在PRINT语句之中完全一样，故不复述。但是有一点要注意，USING给出的格式串是多长，不管该变量字符长短如何，打印时的位置一律按格式串的长度留足。比格式串短的变量值，将被空位占位，小数点后的空位置，将用零来补充。

USING并不对打印位置指定起作用，它只对被打印的变量或数值本身的长短起作用。

例(3) 打印一张1982年6月某河流的日流量表，流量值用INPUT语句由键盘输入。

```
5 CLEAR; CSIZE 3
10 LCURSOR 2 : COLOR 3
20 LPRINT "1982"; "YEAR"
30 INPUT "MONTH"; M
40 A $ = "MONTH", B $ = "DAY"
45 CSIZE 2
50 LPRINT A $; TAB 7; B $; TAB 14; "Q"
55 COLOR 0
60 FOR I= 1 TO 30
70 INPUT "Q="; Q
80 LPRINT TAB 1; M; TAB 7; I; TAB 11; USING
   "#####"; Q
85 USING
90 IF (I-INT (I/5) * 5) = 0 LPRINT
100 NEXT I
120 END
```

程序中第20句、第50句打印表头格式。第70句输入流量Q的值，第80句打印结果。

第90句是为了输出清晰，设计成每打印5行空一行的语句。

4、送纸指令 LF

CE—150上除用按进纸键可以进纸外，还有进纸指令可以用来完成进纸与退纸的动作，这个命令就是LF指令。

一般形式:

(语句号)LF (a)

也可用键盘命令形式送入执行。

其中 (a) 可为正数或负数。正为进纸, 负为退纸, (a) 的含义是指送纸行数。例如 LF2 表示进纸两行, LF - 3 表示退纸 3 行。

进纸或退纸的行数的长短, 是决定于字体大小指令的情况。例如在 CSIZE 1 时, LF 1 进纸一行的长度是 2 毫米; 又如在 CSIZE 3 时, LF 1 的进纸长是 9 毫米。下面表 5-2 给出了它们在不同 CSIZE 指令情况的送纸情况。表 5-2 的第三行表示在不同 CSIZE 情况下的最多退纸行数。因为机器规定最多退纸不得超过 10.24 厘米。

不同字体大小的进退纸数表

表 5-2

CSIZE(字体号)	1	2	3	4	5	6	7	8	9
进纸一行长度(mm)	2	4	6	8	10	12	14	16	18
最多退纸行数	-50	-24	-16	-11	-9	-7	-6	-5	-4

注意, LF 命令只能在 TEXT 方式下执行。

表 5-2 中退纸行数所列的是一次最多退的行数。除一次性退纸不得超过以外 (超过显示 ERROP 71), 多次连续退纸的总行数, 也不得超过表 5-2 中所列的数。

例 (4) 打印一张自然数 31~50 的平方根、平方, 立方的数表的程序

```
10 CSIZE 1:COLOR 3
20 LPRINT TAB 23; 2; TAB 33; 3
25 CSIZE 2
30 LPRINT TAB 2; "N"; TAB 6; "√N";
    TAB 11; "N"; TAB 16; "N"
35 CSIZE 1:COLOR 0
40 FOR N=31 TO 50
45 N1=√N, N1=INT(N1*1E4)/1E4
50 N2=N*N, N3=N2*N
55 LPRINT TAB 3; N; TAB 9; N1;
    TAB 20; N2; TAB 28; N3
60 IF N=INT(N/5)*5 LF1
65 NEXT N:END
```

这个程序运行后将打印一张数表, 请注意打印的数都是左端对齐的。表头的打印也是有趣的。通过它的安排, 可以看出 TAB 函数给我们打印格式设计提供了方便。

另外须注意一点, 在这个程序中没有使用 USING 函数, 输出打印的形式与例 (3) 有什么不一样。例 (3) 中是使用了 USING 函数的。

图 5-1 是本例打印出的格式。

例(5) 以TAB函数用*号打印正弦曲线从0到 2π 的图形的程序

```

10 CLEAR : CSIZE 1
20 FOR I= 0 TO 360 STEP 15
30 X=10 * SIN I : X=INT (X+0.5)
40 LPRINT TAB 18+X, " * "
50 NEXT I
60 CSIZE 2 : END

```

程序运行时注意角度方式应为DEG方式。打印的图形见图5-2。

N	\sqrt{N}	N^2	N^3
31	5.5677	981	29791
32	5.6569	1024	32768
33	5.7446	1089	35937
34	5.8309	1156	39304
35	5.916	1225	42875
36	6	1296	46656
37	6.0827	1369	50653
38	6.1644	1444	54872
39	6.2449	1521	59319
40	6.3245	1600	64000
41	6.4031	1681	68921
42	6.4607	1764	74088
43	6.5574	1849	79507
44	6.6332	1936	85184
45	6.7082	2025	91125
46	6.7823	2116	97336
47	6.8556	2209	103823
48	6.9282	2304	110592
49	7	2401	117649
50	7.071	2500	125000

图5-1



图5-2

5、LLIST指令

LLIST指令可在PRO或RUN方式下用键盘命令形式送入机器执行,也可编入程序执行。

LLIST 命令的功能,是指令机器将已输入机器的源程序在打印机上打印出来。(有些BASIC版本称之为列清单或称卸源)。

在PC-1500机上, LLIST指令形式主要有下面几种:

形式1: LLIST

功能: 从头至尾将源程序全部打印出来。

形式2: LLIST<语句号n>

功能: 把第n号语句打印出来。

形式3: LLIST, <语句号n>

功能: 打印从头开始到n号语句为止的程序。

形式4: LLIST<语句号n>,

功能: 打印从n号语句开始到最后的程序。

形式5: LLIST<语句号m>, <语句号n>

功能: 打印从m号开始到n号结束的所有语句。

除以上五种形式以外, 还可以将语句号改为标号, (如果程序中有标号的话), 如

LLIST"A"、LLIST"X"，但若程序中没有该标号的话，将产生ERROR 11。

LLIST的执行，只能1号或2号两种大小的字体。同其他一些语句命令一样，在显示屏上写LLIST时，不能用在LIST前加L的方法，而只能一次写出。

第二节 绘图语句及指令

在PC-1500系统中，绘图方式命令GRAPH指定以后，CE-150将建立起绘图的X—Y坐标系。在这个坐标系中，可用GLCURSOR与SORGN指令建立起不同的坐标原点，可用LINE与RLINE描线条指令绘出各种图形，同时可用ROTATE对打印字符方向转动，加上前面所介绍的LPRINT指令、COLOR指令、CSIZE指令等等，于是可在打印纸上绘出一些简单图形。

因为CE-150不是专用的绘图仪，所以以上命令的执行被打印机本身的条件所限制，也就是不可能绘大的图形。下面就这些绘图指令的用法及一些规定作些一般介绍。

一、CE-150的X—Y坐标系

CE-150的X—Y坐标系如图5—3所示。在不另给定原点选择命令时，它即以左下角（停机时笔停的位置）为坐标原点。X为横坐标，Y为纵坐标；箭头方向（我们习惯上的数学坐标方向）为正，反方向为负。

CE-150在绘图时，X方向的移动靠笔的移动完成，而Y方向是靠纸的进退来相对完成的。在X方向，最多只能移动43.2毫米；而在Y方向由于机器规定退纸长最多为10.24厘米，故只能移动在102.4毫米以内。因而CE-150绘图的图幅大小不得超过43.2×102.4毫米。

CE-150 X—Y坐标系的单位我们叫它步。也就是笔向X或Y方向移动的最小距离。一步的长度叫步距，CE-150的步距为0.2毫米。我们可以很容易算得，X方向的行程，最多为43.2毫米/0.2毫




图 5—3

米 = 216步；同理Y方向的行程最多为512步。

上面所说的X—Y坐标系只是机器的坐标系，而用户在绘图时总是有自己的坐标系及不同的比例尺的。在绘图时，首先要将用户坐标系与机器坐标系进行变换，也就是说应选择一个合适的坐标比例与坐标原点位置，这是十分重要的。因为有关这方面的问题已不在本讲义范围之内，请参考其他有关书籍，仅在本章的绘图举例中，作了一些简要的介绍。

二、有关绘图的语句指令

1、GLCURSOR与SORGN指令

GLCURSOR的一般形式：

〈语句号〉GLCURSOR〈表达式1，表达式2

其中：〈表达式1〉与〈表达式2〉的值应在-2048到+2047之间。

功能：移笔指令。表示将笔以原点向X和Y方向分别移动〈表达式1〉与〈表达式2〉的步数。

例如：GLCURSOR (30, 50) 表示将笔移到X为30, Y为50的位置上, (见图5-4)。

SORGN指令的一般形式:

〈语句号〉SORGN

功能：它指定笔的现有位置为新的坐标原点。所以它一般跟在移笔指令语句GLCURSOR或描线命令LINE之后使用。例如语句:

GLCURSOR (100, 100) ; SORGN

表示将笔移到坐标(100, 100)作为新的坐标系X'-Y'的原点。见图5-4。

GLCURSOR与SORGN指令, 常常在一起共同选择原点。

2、LINE语句指令 (GRAPH方式下)

LINE描线语句的一般形式:

形式1: 〈语句号〉LINE (X1, Y1) - (X2, Y2), 〈线型号〉, 〈颜色号〉

其中: 〈线型号〉为0~9, 分别表示十种不同线型, 详见表5-3。〈颜色号〉为0~3即黑、蓝、绿、红四色。




图5-4

线条情况表

表5-3

线型号	线条情况	线条图例示意
0	实线	—————
1	0.4mm虚线	-----
2	0.6mm虚线	-----
3	0.8mm虚线	-----
4	1.0mm虚线	-----
5	1.2mm虚线	-----
6	1.4mm虚线	-----
7	1.6mm虚线	-----
8	1.8mm虚线	-----
9	空线(不画线)	

功能：表示用给出的线型与颜色, 从座标点(X1, Y1)描绘一条线到座标点(X2, Y2)。

形式2: 〈语句号〉LINE (X1, Y1) - (X2, Y2) - ... - (X7, Y7), 〈线型号〉, 〈颜色号〉

功能：与形式1同。只是它表示从(X1, Y1)顺序连续描线到(X7, Y7)为止。

例(1) 下面程序描线如图5-5

```
10 GRAPH: SORGN
20 LINE (100, 100)
   - (200, 50), 8, 1
30 TEXT: END
```

例(2) 10 GRARH

```
2) LINE(10,10) - (10,
50) - (50, 50)
- (70, 70) - (70,
```




图 5-5




图 5-6

```
0) - (100, 50) - (100, 100), 0, 0
30 LINE- (150, 100) - (150, 0)
40 TEXT
```

例(2)中, 20句连续描了6段线, 第30句接着又描 2 段。注意30句的第一个坐标点前有“-”号, 表示由前面接着向下描。例(2)绘的图见图 5-6

形式 3: <语句号>LINE (X1, Y1) - (X2, Y2), <线型号>, <颜色号>, B

其中: B表示画一框形的命令。形式 3 表示用给出的线型及颜色, 画一个对角线顶点坐标为 (X1, Y1) 与 (X2, Y2) 的矩形。

```
例(3) 10 GRAPH
20 LINE(100,150)-(200,200),
4, 0, B
30 LINE(125, -25)-(200, 25),
0, 1, B
40 TEXT
50 END
```

运行后, 绘出一个顶点坐标为 (100, 150) 与 (200, 200) 的一个虚线矩形及一个顶点为 (125, -25) 与 (200, 25) 的实线矩形。见(图 5-7)。




图 5-7

关于LINE指令，有几点说明：

①X、Y的取值范围为-2048~+2047。超过此范围显示ERROR 70，但是基于CE-150的特性，X方向不要超过216步。Y方向最大到最小座标不要超过512步。如果超过，笔就停在纸的边缘，无法绘出正确的图来。

②选择原点要注意合适。如果程序没有选原点指令，如例（3），则机器以开机时笔的停留位置为座标原点。

③在LINE语句的三种使用形式中，线型号、颜色号及矩形标志B，均可省略。若省去线型号机器将自动按实线绘图；若省去颜色号则机器将按现在用的色笔来绘图；也可全部省去。例如下面语句省去线型号与颜色号绘矩形图的写法是：

LINE(X1, Y1)-(X2, Y2), , , B
这里省略了线型号 ↑ ↑ 省略了颜色号

又如省去线型号与B的写法是：

LINE (X1, Y1)-(X2, Y2) , , 3
这里省略了线型号 ↑

不过省略的写法要注意逗号“，”不能搞错或省略。总之机器是将第一个逗号后的数认作是线型号，第二个逗号后的数认作颜色号的。

④LINE命令还有一种用法形式就是：

LINE - (X1, Y1)

这种格式的功能是使笔从停笔的某位置一直画到(X1, Y1)。这种用法有其方便之处，也是经常用的。

⑤LINE命令一定要在GRAPH方式下执行，而且注意绘图完毕也应恢复TEXT方式。

3. RLINE描线指令：

RLINE在形式上与功能上都与LINE相类似，不同的是：LINE是相对于原点描线，RLINE是相对于现有笔的位置描线，换句话说，LINE的座标原点是固定的，RLINE的座标系原点在不断的变动。例如它从A点到B点到C点描线，开始以起笔位置为原点，描到点后再向B点描时，又以A点为新的原点，到B点后又以B点为新的原点，这样一直递推下去。

例如我们给出一条命令：

RLINE - (50,50)-(50,-100)-(-50,-50)

执行结果是：笔从起笔位置开始，向X方向走50步，向Y方向走50步；接着向X方向走50步，向Y方向走-100步，又接着向X方向走-50步向Y方向走-50步。(图5-8)

RLINE指令在执行中原点是不断变化的，很




图5-8

容易产生累积误差，这一点要特别注意。


4、ROTATE 打印字符方向指令：（在GRAPH方式下使用）

在图形绘制中，有时需要加上文字，并且是不同方向书写的文字。ROTATE 指令提供了这种可能。

ROTATE语句的一般形式：

〈语句号〉ROTATE〈数〉

其中：〈数〉的值为0~3的正整数，它们分别代表四个不同的方向如下所示。



例（5）简单的打印方向示意的程序

```
10 GRAPH
20 GLCURSOR(50,0):SORGN
30 FOR I=0 TO 3
40 ROTATE I
50 LPRINT"ASDF485"
60 NEXT I
70 TEXT:END
```

大家可以在机器上运行看看，打印的结果如何，这里不画它了。

第三节 CE—150绘图举例

CE—150打印机是一窄行打印机，它虽具有绘制图形的功能，但毕竟不是绘图仪，其绘图功能是极其有限的。尽管如此，学习并掌握绘图的基本方法，对我们以后学习用绘图仪绘图是非常重要的和有益的，下面介绍几个绘图方法的例子。

例（1）绘制一个方格图，每行每列均为十格，每小格的宽和高分别为4mm与6mm。

如前所述，CE—150的纸宽57mm，而有效绘图宽为43.2mm；同时因退纸行程限制在10.24cm，所以图幅最大不得超过43.2×102.4mm，我们已知CE—150的步距为0.2mm，那么它在X方向的最大步数不能超过216步，在Y方向不能超过512步。

根据题意，每小格为4×6mm，也就是每一格在X方向走20步，Y方向走30步（因为4/0.2=20，6/0.2=30）。编制程序如下：

```
10 CLEAR
20 GRAPH
30 GLCURSOR(0,0):SORGN
40 FOR K=0 TO 200 STEP20
50 LINE (K, 0) - (K, -300), 0, 2
60 NEXT K
70 FOR I=0 TO -300 STEP -30
80 LINE (0, I) - (200, I)
90 NEXT I
```

100 TEXT : END

程序中30语句指定原点，第40句至60句绘纵线。第70至90句绘横线。第100句恢复TEXT方式。在第50句中指定用了绿色实线绘图。

绘出图形见图5—9。为每小格均为4×6mm每行每列都是10格。

例(2) 绘制一个半径为R的圆的方法，设R=15mm。

这个例题，如用圆的参数方程来作非常方便。圆的坐标参数方程为：

$$\begin{cases} X = R * \cos E \\ Y = R * \sin E \end{cases}$$

可以看出，根据这个参数方程，角度E从0~360°旋转一周，便可以得到圆的轨迹，(图5—10)。于是剩下的问题是选择坐标原点与确定E的变化增值量。我们根据机器坐标系统的规定，选择原点为(100, -100)，E的增量暂定3°，估计能满足绘图的精度要求。

程序编制如下，半径R由INPUT语句输入。

```

10 GRAPH
20 INPUT "R="; R : R=R/0.2
30 GLCURSOR (110, -110) :
   SORGN
40 P=R, Q=0
50 FOR E=0 TO 360 STEP 3
60 X=R * COS E,
   Y=R * SIN E
70 LINE (P, Q) - (X, Y)
75 P=X, Q=Y
80 NEXT E
90 TEXT : END

```

此题在运行时要注意角度使用方式为DEG，不能搞错。

例(3) 根据实测资料，某河某断面的水位流量关系如下表所列。绘出它的水位流量关系曲线。

水位(米)	360.0	366.8	373.0	381.3	387.0	391.5	395.6	400.0
流量(秒立米)	1000	2500	5000	10000	15000	20000	25000	30000




图5—9




图5—10

这个题目，是将一组离散点绘成一条尽可能较为光滑的曲线。主要关键是：①进行比例变换，使用户座标与机器座标系统一。②选择一个合适的数学方法来拟合离散点。第一步，进行比例变换。

根据题目，设水位为Z，流量为Q，那么：

$$Z_{\max} = 400 \quad Z_{\min} = 360 \quad \Delta Z = 40 \text{ (米)}$$

$$Q_{\max} = 30000 \quad Q_{\min} = 1000 \quad \Delta Q = 29000$$

为了方便，干脆取 $Q_{\min} = 0$ ，则 $\Delta Q = 30000$ 秒立米。

根据CE—150打印纸的情况，设X方向为Z座标，Y方向为Q座标，我们已知步距为0.2mm。设X轴1mm为1米， $\therefore \Delta Z = 40$ 那么图宽为40mm < 43.2mm。

$$LX = 40\text{mm}/0.2 = 200\text{步}$$

$$KX = 200/40 = 5$$

LX—CE—150座标系的X方向最大步数（亦即图幅宽）。

KX—X方向比例尺。

又设Y轴3mm为1千秒立米， $\therefore \Delta Q = 30000$ ，图长为90mm < 10.24cm

$$LY = 90/0.2 = 450\text{步}$$

$$KY = 450/30000 = 0.015$$

LY—Y方向最大步数。

KY—Y方向比例尺。

第二步，绘出上面所计算出的座标格（程序的“B”标号一段，从45句到125句）。为观看方便，座标格用红色。取KY为负值，是为了将Y座标反向，以使曲线座标适合平常我们的习惯方向。X座标每5米画一条线，Y座标每2000秒立米画一条线。

第三步，选取适线方法。考虑到精度可以满足要求，所以在这里采用拉格朗日插值法进行计算，并将它作为子程序形式放在程序之中。根据图形的比例情况，我们每取水位变化1米计算1个流量值。

标号“A”一段是输入水位流量实测数据，数据放在DATA语句之中；标号“B”一段是绘座标格程序；标号“Z”一段是绘关系曲线程序。程序如下：

```

1  "A" CLEAR
4  DATA 1000, 360, 2500, 366.8, 5000, 373, 10000, 381.3, 15000,
      387, 20000, 391.5
6  DATA 25000, 395.6, 30000, 400
10 INPUT "J="; J
15 DIM X (J), Y (J)
20 FOR D=0 TO J
30 READ Y (D), X (D)
35 NEXT D
40 KX = 200 / (X (J) - X (0)),
      KY = 450 / Y (J) : END
45  "B" GRAPH : CSIZE 1

```



```

50 GLCURSOR (10, 0) : SORGN : COLOR 3
55 GLCURSOR (0, 30) : ROTATE 1
60 FOR XI=0 TO KX*(X(J)-X(0)) STEP KX*5
65 LPRINT XI/KX+360
70 LINE (XI, 0) - (XI, -450), 0, 3
75 LINE - (XI+KX*5, 30), 9
80 NEXT XI
90 FOR YI=0 TO -KY*Y(J) STEP -KY*2000
95 LINE (0, YI) - (200, YI), 0, 3
100 ROTATE 1
105 LINE - (-10, YI+10), 9
110 LPRINT (-YI/KY)/1000
115 NEXT YI : LINE - (-10, -470), 9
120 LPRINT "Q*1000"
125 LINE - (0, 0), 9 : END
130 "Z" FOR X=X(0) TO X(J) STEP 1
135 GOSUB 500
140 XI=(X-X(0))*KX, YI=-Y*KY
145 LINE (Z, Q) - (XI, YI), 0, 0
150 Z=XI, Q=YI
155 NEXT X : TEXT : END
500 I=0
510 IF X>=X(I) AND X<=X(I+1) THEN 600
520 I=I+1
530 IF I>=J-2 LET I=J-2 : GOTO 600
540 GOTO 510
600 A=Y(I)*(X-X(I+1))*(X-X(I+2))/(X(I)-X(I+1))/
(X(I)-X(I+2))
610 B=Y(I+1)*(X-X(I))*(X-X(I+2))/(X(I+1)-X(I))/
(X(I+1)-X(I+2))
620 C=Y(I+2)*(X-X(I))*(X-X(I+1))/(X(I+2)-X(I))/
(X(I+2)-X(I+1))
630 Y=A+B+C : Y=INT Y
640 RETURN

```

程序运行后，绘出Z~Q曲线如图5—11所示。程序运行要用标号“A”，“B”，“Z”启动三次，“A”标号启动后，J的值输入7。

图5—11所标纵座标为水位高程Z，横座标流量Q，其单位应乘1000。




图 5—11

例（4）绘制三相交流电电动势波形图

本例只画一个周期的图形，三相交流电的A相、B相、C相分别用蓝、绿、红三种不同颜色画出。程序如下：

```

5 CLEAR : LF 6 : GRAPH
10 GLCURSOR ( 5 , 0 ) : SORGN
15 LINE ( 0 , 0 ) - ( 210 , 0 ) , , 0
20 RLINE - ( -10 , 2 ) - ( 0 , 4 ) - ( 10 , 2 )
25 GLCURSOR ( 190 , 5 ) : LPRINT "2π"
30 LINE ( 0 , -120 ) - ( 0 , 120 )
35 RLINE - ( 2 , -10 ) - ( -4 , 0 ) - ( 2 , 10 )
40 LPRINT "Y"
45 LINE ( 100 , -2 ) - ( 100 , 2 ) : LPRINT "π" :
   GLCURSOR ( 0 , 0 )
50 FOR K = 0 TO 200 STEP 3
55 A = 100 * SIN ( K * 1.8 )
60 LINE - ( K , A ) , , 1
65 NEXT K : GLCURSOR ( 0 , 0 )
100 FOR K = 0 TO 200 STEP 3
110 B = 100 * SIN ( K * 1.8 - 120 )
120 LINE - ( K , B ) , , 2
130 NEXT K : GLCURSOR ( 0 , 0 )
200 FOR K = 0 TO 200 STEP 3
210 C = 100 * SIN ( K * 1.8 - 240 )
220 LINE - ( K , C ) , , 3
230 NEXT K
240 TEXT : LF 10
250 END

```

程序运行后后，绘制的图形如图 5—12 所示。这个程序中，多次用到了 RLINE 命令。

10~45语句是绘座标,用黑色。在这一段中注意GLCURSOR的用途与用法。50~65语句是绘A相的正弦波形,100~130语句是绘B相的正弦波形,140~230语句是绘C相的正弦波形。实际绘出的图是彩色的。




图 5—12




图 5—13

例 (5) 用图解法解联立方程组

$$\begin{cases} 3y - x^2 + 6x = -12 \cdots \cdots (1) \\ 2y - x = 4 \cdots \cdots (2) \end{cases}$$

图解法解方程组就是用解析几何的方法求两方程几何轨迹的交点。方程 (1) 是抛物线方程,方程 (2) 是直线方程。若有一个交点就有一组解,若有二个交点则有二组解;若无交点,就无实数解。

首先,我们应大致确定选取的笛卡儿坐标系取数范围,以便确定图幅的比例。从分析可知,X轴从-4到10,Y轴从-8到12基本能满足方程图形所需。

第二步进行比例转换。因为X取-4~10,所以 $\Delta X = 14$,而 $\Delta Y = 20$ 。为统一起见,两比例均以X比例尺为准。考虑图幅X宽40mm,故:

$$LX = 40/0.2 = 200 \text{步}, K = 200/\Delta X = 200/14$$

第三步选取座标原点后绘图。座标原点只要能使X从-4到10,Y从-8到12均可。

程序如下,对座标轴及原点的换算简化了一些。程序中:

第15语句:建立座标原点。

第20~85语句:绘笛卡儿座标图,用黑色。

第110~175语句:描绘方程(1)的图形,用红色。

第200~290语句:描绘方程(2)的图形,用绿色绘。

```
5 CLEAR : K = 200/14
10 GRAPH : CSIZE 2
15 GLCURSOR (60, ~160) : SORGN
20 LINE (-60, 0) - (140, 0), , 0
```

```

25 RLINE - (-10, 2) - (0, -4) - (10, 2)
30 LPRINT "X"
40 LINE (0, -120) - (0, 180)
45 RLINE - (-2, -10) - (4, 0) - (-2, 10)
50 LPRINT "Y": CSIZE 1
55 FOR X=-4 TO 8 STEP 2
60 LINE (X*K, 0) - (X*K, 5) :LPRINT X
65 NEXT X
70 FOR Y=-8 TO 12 STEP 2
75 LINE (-2, Y*K) - (2, Y*K)
80 IF Y LPRINT Y
85 NEXT Y
100 DIM X (15) , Y (15)
110 FOR I= 1 TO 15
120 X (I) =I- 5
130 Y (I) = (X (I) ^ 2 - 6 * X (I) -12) / 3
140 X (I) =X (I) * K, Y (I) =Y (I) * K
150 NEXT I
155 P=X (1) , Q=Y (1)
160 FOR I= 1 TO 15
165 LINE (P, Q) - (X (I) , Y (I) ) , , 3
170 P=X (I) , Q=Y (I)
175 NEXT I
200 FOR I= 1 TO 15
210 X (I) =I- 5
220 Y (I) = (X (I) + 4) / 2
230 X (I) =X (I) * K, Y (I) =Y (I) * K
240 NEXT I
250 P=X (1) , Q=Y (1)
260 FOR I= 1 TO 15
270 LINE (P, Q) - (X (I) , Y (I) ) , , 2
280 P=X (I) , Q=Y (I)
290 NEXT I : CSIZE 0
300 TEXT : LF 12 : END

```

程序运行后, 描绘如图 5—13 的图形, 可以看出有两个交点, 也就是有两组解。其中一组(近似)解是:

$$\begin{cases} x = -1.9124 \\ y = 1.0438 \end{cases}$$


这组解并不是从图上量出的。因为这个方程组的解不是整数，图形的精度有限，无法精确量出。只是看出其解在 $(-2, 1)$ 附近而已，另一组解请读者自己求解。

习 题 七

一、第三节例(2)若改为绘制一个正六边形，程序如何修改？若改为绘长短半径分别为 3 cm, 5 cm 的椭圆，又如何改？

二、编一程序，绘出一个半径为 18mm，宽为 12mm，圆心角为 $\pi/3.5$ 的扇面形。

三、设计一程序，绘出下图所示悬臂梁的弯矩图。



四、某厂五个车间的产值分别是14万元、23万元、18万元、25万元、16万元。绘一张反映各产值的直方图。